ABSTRACT
Data comes in all forms, shapes, sizes and complexities. Stored in files and data sets, SAS® users across industries know all too well that data can be, and often is, problematic and plagued with a variety of issues. When unique and reliable identifiers are available, users routinely are able to match records from two or more data sets using merge, join, and/or hash programming techniques without problem. But, what happens when a unique identifier, referred to as the key, is not reliable or does not exist. These types of problems are common and are found in files containing a subscriber name, mailing address, and/or misspelled email address, where one or more characters are transposed, or are partially and/or incorrectly recorded? This presentation introduces what fuzzy matching is, a sampling of data issues users have to deal with, popular data cleaning and user-defined validation techniques, the application of the CAT functions, the SOUNDEX (for phonetic matching) algorithm, SPEDIS, COMPGED, and COMPLEV functions, and an assortment of programming techniques to resolve key identifier issues and to successfully merge, join and match less than perfect or messy data.

INTRODUCTION
When data sources and data sets contain consistent and valid data values, share common unique identifier(s), and have no missing data, the matching process rarely presents any problems. But, when data originating from multiple sources contain duplicate observations, duplicate and/or unreliable keys, missing values, invalid values, capitalization and punctuation issues, inconsistent matching variables, and imprecise text identifiers, the matching process is often compromised by unreliable and/or unpredictable results. When issues like these exist, SAS users must first clean and standardize any and all data irregularities before any attempts to match data records are performed. To assist in this time-consuming and costly process, users often utilize special-purpose programming techniques including the application of one or more SAS functions, the use of approximate string matching, and/or an assortment of constructive programming techniques to standardize and combine data sets together.

The examples presented in this paper illustrate two data sets, Movies_with_Messy_Data and Actors_with_Messy_Data. The Movies_with_Messy_Data data set, illustrated in Figure 1, consists of 31 observations, a data structure of six variables where Title, Category, Studio, and Rating are defined as character variables; and Length and Year are defined as numeric variables. After careful inspection several data issues can be found in this data set including the existence of missing data, duplicate observations, spelling errors, punctuation inconsistencies, and invalid values.

The Actors_with_Messy_Data data set, illustrated in Figure 2, contains 15 observations and a data structure consisting of three character variables: Title, Actor_Leading and Actor_Supporting. As with the Movies_with_Messy_Data data set, several data issues are found including missing data, spelling errors, punctuation inconsistencies, and invalid values.
THE MATCHING PROCESS EXPLAINED

In an age of endless spreadsheets, apps and relational database management systems (RDBMS), it’s unusual to find a single sheet, file, table or data set that contains all the data needed to answer an organization’s questions. Today’s data exists in many forms and all too often involves matching two or more data sources to create a combined file. The matching process typically involves combining two or more data sets, spreadsheets and/or files possessing a shared, common and reliable, identifier (or key) to create a single data set, spreadsheet and/or file. The matching process, illustrated in the following diagram, shows two tables with a key, Title, to combine the two tables together.

Figure 1. Movies_with_Messy_Data data set.

Figure 2. Actors_with_Messy_Data data set.
Figure 3. Joining movies to actors.

But, when a shared and reliable key is associated with input data sources that are nonexistent, inexact, or unreliable, the matching process often becomes more involved and problematic. As cited in Sloan and Hoicowitz (2016), special processes are needed to successfully match the names and addresses from different files when they are similar, but not exactly the same. In a constructive and systematic way the authors of this paper describe a six step approach to cleansing data and performing fuzzy matching techniques.

**Step 1: Remove extraneous characters.**
As a general rule, punctuation can differ while the names are the same. For example, John’s “super” pizza and John’s super pizza refer to the same restaurant. Therefore, we remove the following characters from all names: ‘ “ & ? - .

**Step 2: Put all characters in upper-case notation and remove leading blanks.**

**Step 3: Remove words that might or might not appear in the same company name.**
Some examples are The, .com, Inc, LTD, LLC, DIVISION, CORP, CORPORATION, CO., and COMPANY.

**Step 4: Rationalize the zip codes when matching addresses.**
We found it useful to remove the last 4 digits of 9-digit zip codes, because some files might only have 5-digit zip codes. Since some files might have zip codes as numeric fields, and other files might have zip codes as character fields, make sure to include leading zeroes. For example, zip codes with a leading zero, as in 08514, would appear in a numeric field as 8514 requiring the leading zero to be inserted.

If working with US zip codes, make sure they are all numeric. This may not apply for other countries. One common mistake to watch for is that sometimes Canada, with abbreviation CA, is put in as the state CA (California) instead of the country CA. Since Canada has an alphanumeric 6-character zip code, this, hopefully, will be caught when checking for numeric zip codes.

**Step 5: Choose a standard for addresses.**
Decide whether to use Avenue or Ave, Road or Rd, etc, and then convert the address fields to match the standard.

**Step 6: Match the names and addresses using one or more fuzzy matching techniques.**
Users have an assortment of powerful SAS algorithms, functions and programming techniques to choose from.

Fuzzy matching is the process by which data is combined where a known key either does not exist and/or the variable(s) representing the key is/are unreliable. In Dunn (2014), the author suggests addressing these types of scenarios using the following steps.

1. Determine the likely matching variables using metadata (e.g., PROC CONTENTS, etc.) listings.
2. Perform data cleaning.
3. Use the COMPGED function to determine the dissimilarity between two strings.
The authors of this paper agree with Sloan & Hoicowitz, and Dunn’s strategies for handling fuzzy matching issues. But, we also want to stress the importance of understanding the physical side of data along with the distribution of data values. To address these areas, we suggest adhering to a five step approach, as follows:

1. Determine the likely matching variables using metadata (e.g., PROC CONTENTS, etc.) listings.
2. Understand the distribution of data values including the number of levels for categorical and key variables.
3. Perform data cleaning.
4. Perform data transformations.
5. Use Fuzzy matching programming techniques when a reliable key between data sources are nonexistent, inexact or unreliable.

### STEP #1: DETERMINING THE LIKELY MATCHING VARIABLES

This first step determines whether any variables exist for matching purposes. Using a PROC CONTENTS alphabetical list of variables and attributes listing for the data sets, Movies_with_Messy_Data and Actors_with_Messy_Data, shown below; compare each variable assessing the likelihood of potential matching variables. The PROC CONTENTS code is illustrated below.

**PROC CONTENTS Code:**

```plaintext
proc contents data=mydata.Movies_with_Messy_Data ;
  run ;
proc contents data=mydata.Actors_with_Messy_Data ;
  run ;
```

From the PROC CONTENTS listing, illustrated in Figure 4, we see that TITLE is consistently defined in both data sets as a $30 character variable. Based on this, we examine the values of the TITLE variable in greater detail to determine whether it can serve as the key for matching observations in both data sets, as well as the distribution of data values for other categorical variables.

### STEP #2: UNDERSTANDING THE DISTRIBUTION OF DATA VALUES AND NLEVELS

To derive a more accurate picture of the data sources, we suggest that users conduct extensive data analysis by identifying missing values, outliers, invalid values, minimum and maximum values, averages,
value ranges, duplicate observations, distribution of values, and the number of distinct values a categorical variable contains. This important step provides an understanding of the data, while leveraging the data cleaning and standardizing activities that will be performed later. One of the first things data wranglers will want to do is explore the data using the FREQ procedure.

PROC FREQ Code:

```
proc freq data=mydata.Movies_with_Messy_Data ;
    tables Title / NOCUM NOPERCENT
    out=Missing_Titles(where=(Title = "")) ;
run ;
```

Reviewing the FREQ results, we see there are duplicate “key” values and missing values, as shown in Figure 5.

<table>
<thead>
<tr>
<th>Title</th>
<th>Frequency</th>
</tr>
</thead>
<tbody>
<tr>
<td>Brave Heart</td>
<td>2</td>
</tr>
<tr>
<td>Casablanca</td>
<td>1</td>
</tr>
<tr>
<td>Christmas Vacation</td>
<td>1</td>
</tr>
<tr>
<td>Christmas Vacation</td>
<td>1</td>
</tr>
<tr>
<td>Coming to America</td>
<td>1</td>
</tr>
<tr>
<td>Dracula</td>
<td>1</td>
</tr>
<tr>
<td>Dressed to Kill</td>
<td>1</td>
</tr>
<tr>
<td>Forrest Gump</td>
<td>2</td>
</tr>
<tr>
<td>Forrest Gumppp</td>
<td>1</td>
</tr>
<tr>
<td>Ghost</td>
<td>1</td>
</tr>
</tbody>
</table>

| Jaws         | 1         |
| Jurassic Park | 1        |
| Lethal Weapon | 1        |
| Michael      | 1         |
| Micheal      | 1         |
| National Lampoon's Vacation | 1 |
| National Lampoons Vacation | 1 |
| Poltergeist  | 1         |
| Rocky        | 2         |
| Scarface     | 1         |

| Silence of the Lambs | 1 |
| Star Wars            | 1 |
| The Hunt for Red October | 1 |
| The Terminator       | 1 |
| The Wizard of Oz     | 1 |
| The Wizard of Ozz    | 1 |
| Titanic              | 1 |

Figure 5. PROC FREQ Results show duplicate “key” values and missing values.

Determining the number of distinct values a categorical variable has is critical knowledge that all data analysts and wranglers seek an answer to. Acquiring this information helps everyone involved to better understand the number of distinct variable levels, the unique values and the number of occurrences for developing data-driven programming constructs and elements. The FREQ procedure provides details about the number of levels for each categorical variable.

PROC FREQ Code:

```
title "NLevels for Variables of Interest in Movies_with_Messy_Data" ;
proc freq data=mydata.Movies_with_Messy_Data nlevels ;
    tables Title Rating Category Studio / nopct nocum ;
run ;
```

Reviewing the PROC FREQ results, we see the distinct variable levels for each variable: Title, Rating, Category and Studio, as shown in Figure 6.
Reviewing the PROC FREQ results, an assortment of data consistency, validation and capitalization issues have been identified for each variable, as shown in Figure 7.

Figure 7. PROC FREQ results depict unique values and the number of occurrences for each variable of interest.

**STEP #3: PERFORMING DATA CLEANING**

Data cleaning, often referred to as data scrubbing, is the process of identifying and fixing data quality issues including missing values, invalid character and numeric values, outlier values, value ranges, duplicate observations, and other anomalies found in data sets. SAS provides many powerful ways to perform data
cleaning tasks. For anyone wanting a complete guide to the various SAS data cleaning techniques, we highly recommend *Cody's Data Cleaning Techniques Using SAS, Third Edition*. To illustrate one popular data cleaning technique that users frequently turn to for identifying and removing duplicate observations, we illustrate the SORT procedure.

**Exploring PROC SORT to Identify and Remove Duplicate Observations**

A popular approach with users for identifying and removing duplicate observations in a data set is to use PROC SORT. By using the SORT procedure’s three options: **DUPOUT=**, **NODUPRECS**, and **NODUPKEYS**, users are better able to control how duplicate observations are identified and removed.

**Specifying the DUPOUT= Option**

PROC SORT’s **DUPOUT=** option is often used to identify duplicate observations before actually removing them from a data set. A **DUPOUT=** option, often specified when a data set is too large for visual inspection, can be used with the **NODUPKEYS** or **NODUPRECS** options to name a data set that contains duplicate keys or entire observations. In the next example, the **DUPOUT=**, **OUT=** and **NODUPKEY** options are specified to identify duplicate keys.

PROC SORT Code:

```sas
PROC SORT DATA=mydata.Movies_with_Messy_Data
   DUPOUT=Movies_Dupout_NoDupkey
   OUT=Movies_Sorted_Cleaned_NoDupkey
   NODUPKEY ;
   BY Title ;
RUN ;
```

```sas
PROC PRINT DATA=work.Movies_Dupout_NoDupkey NOOBS ;
   TITLE “Observations Slated for Removal” ;
RUN ;
PROC PRINT DATA=work.Movies_Sorted_Cleaned_NoDupkey NOOBS ;
   TITLE “Cleaned Movies Data Set” ;
RUN ;
```

Results:

![Figure 8. Duplicate observations to be removed.](image-url)
Figure 9. The data set after cleansing.

PROC SORT's **NODUPRECS** (or **NODUPREC**) (or **NODUP**) option identifies observations with identical values for all columns. In the next example, the **OUT=**, **DUPOUT=** and **NODUPRECS** options are specified. Note that fewer observations are removed than with **NODUPKEY**, which removes for duplicates in specific columns.

**PROC SORT Code:**

```sql
PROC SORT DATA=mydata.Movies_with_Messy_Data
    DUPOUT=Movies_Dupout_NoDupRecs
    OUT=Movies_Sorted_Cleaned_NoDupRecs
    NODUPRECS;
    BY Title;
RUN;
```

```sql
PROC PRINT DATA=work.Movies_Dupout_NoDupRecs NOOBS;
    TITLE "Observations Slated for Removal";
RUN;
```

```sql
PROC PRINT DATA=work.Movies_Sorted_Cleaned_NoDupRecs NOOBS;
    TITLE "Cleaned Movies Data Set";
RUN;
```
Results:

![Observations Stated for Removal](image1)

Figure 10. Duplicate observations to be removed.

![Cleaned Movies Data Set](image2)

Figure 11. The data set after cleansing.

**Note:** Although the removal of duplicates using PROC SORT is a popular technique among many SAS users, an element of care should be given to using this method when processing large data sets. Since sort operations can often be CPU-intensive operations, the authors of this paper recommend comparing PROC SORT to procedures like PROC SUMMARY with the CLASS statement to determine the performance impact of one method versus another.
EXPLORING SAS FUNCTIONS TO MODIFY DATA

SAS functions are an essential component of the SAS Base software. Representing a variety of built-in and callable routines, functions serve as the “work horses” in the SAS software providing users with “ready-to-use” tools designed to ease the burden of writing and testing often lengthy and complex code for a variety of programming tasks. The advantage of using SAS functions is evident by their relative ease of use, and their ability to provide a more efficient, robust and scalable approach to simplifying a process or programming task.

SAS functions span a number of functional categories, including character, numeric, character string matching, data concatenation, truncation, data transformation, search, date and time, arithmetic and trigonometric, hyperbolic, state and zip code, macro, random number, statistical and probability, financial, SAS file I/O, external files, external routines, sort, to name a few. The next example illustrates an old, an alternate, and new way of concatenating strings and/or variables together. The code, results and analysis appear below.

DATA Step and CAT Functions:

data _null_ ;
  length NUM 3. A B C D E $ 8 BLANK $ 1 ;
  A = 'The' ;
  NUM = 5 ;
  B = ' Cats' ;
  C = 'in' ;
  D = ' the' ;
  E = 'Hat' ;
  BLANK = ' ' ;

* Old way of concatenating with TRIM and LEFT functions and concatenation operator ;
 OLD = trim(left(A)) || BLANK || trim(left(NUM)) || BLANK || trim(left(B)) ||
       BLANK || trim(left(C)) || BLANK || trim(left(D)) || BLANK || trim(left(E)) ;

* Using the STRIP function and concatenation operator ;
 STRIP = strip(A) || BLANK || strip(NUM) || BLANK || strip(B) || BLANK ||
       strip(C) || BLANK || strip(D) || BLANK || strip(E) ;

* Using the CAT functions to concatenate character and numeric values together ;

1 CAT = cat (A, NUM, B, C, D, E) ;
2 CATQ = catq(BLANK, A, NUM, B, C, D, E) ;
3 CATS = cats(A, NUM, B, C, D, E) ;
4 CATT = catt(A, NUM, B, C, D, E) ;
5 CATX = catx(BLANK, A, NUM, B, C, D, E) ;
   put OLD= / STRIP= / CAT= / CATQ= / CATS= / CATT= / CATX= / ;
run ;
Results:

OLD=The 5 Cats in the Hat
STRIP=The 5 Cats in the Hat
CAT=The 5 Cats in the Hat
CATQ="The 5 Cats in the Hat"
CATS=The 5 Cats in the Hat
CATT=The 5 Cats in the Hat
CATX=The 5 Cats in the Hat

Analysis:
In the preceding SAS code, a single numeric variable, NUM, and six character variables: A, B, C, D, E, and BLANK are defined with their respective values as: NUM=5, A='The', B=' Cats', C='in', D=' the', E='Hat' and BLANK=' '. The oldest way of concatenating two or more strings or variables together is specified using the TRIM and LEFT functions and the concatenation operator "||" in an assignment statement. An alternate approach uses a STRIP function with the concatenation operator "||" in an assignment statement to join two or more strings or variables together. Finally, the newer and more robust concatenation approach uses the CAT family of functions: CAT, CATQ, CATS, CATT, and CATX.

1) CAT, the simplest of concatenation functions, joins two or more strings and/or variables together, end-to-end producing the same results as with the concatenation (double bar) operator.

2) CATQ is similar to the default features of the CATX function, but the CATQ function adds quotation marks to any concatenated string or variable.

3) CATS removes all leading and trailing blanks and concatenates two or more strings and/or variables together.

4) CATT removes trailing blanks and concatenates two or more strings and/or variables together.

5) CATX, perhaps the most robust CAT function, removes leading and trailing blanks and concatenates two or more strings and/or variables together with a delimiter between each.

VALIDATING DATA WITH PROC FORMAT
Problems with data often necessitate time-consuming validation activities. The strategy is to take the time to become familiar with the data and to discover any problems before expending data analysis and reporting resources. A popular technique used by many to identify data issues is to use the FORMAT procedure. In the next example, a user-defined format is created with PROC FORMAT, a DATA step identifies data issues associated with the Category variable, and a PROC PRINT is specified to display the Category variable’s data issues.

PROC FORMAT, DATA Step and PROC PRINT Code:

```
PROC FORMAT LIBRARY=WORK ;
VALUE $Category_Validation
   'Action' = 'Action'
   'Action Adventure' = 'Action Adventure'
   'Action Cops & Robber' = 'Action Cops & Robber'
   'Action Sci-Fi' = 'Action Sci-Fi'
   'Adventure' = 'Adventure'
   'Comedy' = 'Comedy'
```
'Drama'                = 'Drama'
'Drama Mysteries'      = 'Drama Mysteries'
'Drama Romance'        = 'Drama Romance'
'Drama Suspense'       = 'Drama Suspense'
'Horror'               = 'Horror'
Other                  = 'ERROR - Invalid Category'

RUN;

DATA Validate_Category;
  SET mydata.Movies_with_Messy_Data;
  Check_Category = PUT(Category,$Category_Validation.);
  IF Check_Category = 'ERROR - Invalid Category' THEN
    DO;
      PUT 'Category Error: ' Title;
      OUTPUT;
    END;
  RUN;

PROC PRINT DATA=work.Validate_Category
  NOOBS
  N;
  TITLE "Validation Report for Movie Category Variable";
  VAR Category Title Rating Length Studio Year;
RUN;

SAS Log:
The error messages for the variable, Check_Category, are displayed, below.

  Category Error: Brave Heart
  Category Error: Titanic
  Category Error: Forrest Gump
  Category Error: Christmas Vacation
  Category Error:

Results:

<table>
<thead>
<tr>
<th>Category</th>
<th>Title</th>
<th>Rating</th>
<th>Length</th>
<th>Studio</th>
<th>Year</th>
</tr>
</thead>
<tbody>
<tr>
<td>Action Adventure</td>
<td>Brave Heart</td>
<td>R</td>
<td>177</td>
<td>Paramount Pictures</td>
<td>1995</td>
</tr>
<tr>
<td>Drama Romance</td>
<td>Titanic</td>
<td>PG-13</td>
<td>194</td>
<td>Paramount Pictures</td>
<td>1997</td>
</tr>
<tr>
<td>Drama</td>
<td>Forrest Gump</td>
<td>PG-13</td>
<td>143</td>
<td>Paramount Pictures</td>
<td>1994</td>
</tr>
<tr>
<td>Comedy</td>
<td>Christmas Vacation</td>
<td>PG-13</td>
<td>97</td>
<td>Warner Brothers</td>
<td>1989</td>
</tr>
<tr>
<td>Action Adventure</td>
<td></td>
<td>R</td>
<td>177</td>
<td>Paramount Pictures</td>
<td>1995</td>
</tr>
</tbody>
</table>

N = 5
STEP #4: PERFORMING DATA TRANSFORMATIONS

Data transformations are frequently performed by SAS users. From converting a data set structure from wide to long, long to wide, observations to variables, variables to observations, and more, SAS users have a number of choices available to them. A popular procedure used to transform selected variables into observations and observations into variables is the TRANSPOSE procedure. Although PROC TRANSPOSE isn’t designed to print or display output, it is handy for restructuring data in a data set, and is typically used in preparation for special types of processing such as, array processing. In its simplest form, data can be transformed with, or without, grouping. In the example, below, an ungrouped transformation is performed on only the numeric variables in the data set.

PROC TRANSPOSE Code:

```
PROC TRANSPOSE DATA=mydata.Movies_with_Messy_Data
   OUT=Movies_Transposed ;
RUN ;
```

Results:

![Figure 13. Output from PROC TRANSPOSE.](image)

Data can be restructured with PROC TRANSPOSE using a grouping variable. In the next example, the Movies data set is first sorted in ascending order by the variable RATING, the sort results written to the Movies_Sorted data set, and then the Movies_Sorted data set is transposed using the RATING variable as the by-group variable.

PROC TRANSPOSE Code:

```
PROC SORT DATA=mydata.Movies_with_Messy_Data
   OUT=Movies_Sorted ;
   BY Rating ; /* BY-Group to Transpose */
RUN ;

PROC TRANSPOSE DATA=work.Movies_Sorted
   OUT=Movies_Transposed ;
   VAR Title ; /* Variable to Transpose */
   BY Rating ; /* BY-Group to Transpose */
RUN ;

PROC PRINT DATA=Movies_Transposed ;
RUN ;
```

Results:
Figure 14. Output from PROC PRINT after PROC TRANSPOSE

Step #5: Using Fuzzy Matching Programming Techniques
Fuzzy matching is an essential programming technique used by organizations every day, particularly when the matching variables between data sets are non-existent or unreliable. Although this type of processing can be more involved than traditional matching processing techniques (e.g., interleaving, match-merging, joining, etc.), SAS users have a number of powerful functions available to them, including the Soundex (phonetic matching) algorithm, and the SPEDIS, COMPGED and COMPLEV functions, to help make fuzzy matching easier and more effective to use.

Exploring the Soundex Algorithm
The Soundex (phonetic matching) algorithm involves matching files on words that sound alike. As one of the earliest fuzzy matching techniques, Soundex was invented and patented by Margaret K. Odell and Robert C. Russell in 1918 and 1922 to help match surnames that sound alike. It is limited to finding phonetic matches and adheres to the following rules when performing a search:

- Is case insensitive (ignores case);
- Ignores embedded blanks and punctuations;
- Is better at finding English-sounding names.

Although the Soundex algorithm does a fairly good job with English-sounding names, it often falls short when dealing with non-English sounding names. In Foley (1999) the author corroborates this by stating, “The Soundex algorithm is not infallible since it has been known to miss similar-sounding surnames like Rogers and Rodgers while matching dissimilar surnames such as Hilbert and Heibronn.”

So, how does the Soundex algorithm work? As implemented, SAS determines whether a name (or a variable’s contents) sounds like another by converting each word to a code. The value assigned to the code consists of the first letter in the word followed by one or more digits. Vowels, A, E, I, O and U, along with H, W, Y, and non-alphabetical characters do not receive a coded value and are ignored; and double letters (e.g., ‘TT’) are assigned a single code value for both letters. The codes derived from each word conform to the letters and values found in the table, below.

<table>
<thead>
<tr>
<th>Letter</th>
<th>Value</th>
</tr>
</thead>
<tbody>
<tr>
<td>B, P, F, V</td>
<td>1</td>
</tr>
<tr>
<td>C, S, G, J, K, Q, X, Z</td>
<td>2</td>
</tr>
<tr>
<td>D, T</td>
<td>3</td>
</tr>
<tr>
<td>L</td>
<td>4</td>
</tr>
<tr>
<td>M, N</td>
<td>5</td>
</tr>
<tr>
<td>R</td>
<td>6</td>
</tr>
</tbody>
</table>

Figure 15. Soundex code values.

To examine how the movie title, Rocky, is assigned a value of R22, R has a value of 6 but is retained as R, O is ignored, C is assigned a value of 2, K is assigned a value of 2, and Y is ignored. The converted code for
“Rocky” is then matched with any other name that has the same assigned code. The general syntax of the Soundex algorithm takes the form of: Variable =* “character-string”.

In the next example, the Soundex algorithm is illustrated using the =* operator in a simple DATA step WHERE statement and a PROC SQL WHERE-clause to find similar sounding Movie Titles.

Soundex (=*) Algorithm:

```plaintext
DATA Soundex_Matches ;
   SET mydata.Movies_with_Messy_Data ;
   WHERE Title =* "Michael" ;
   RUN ;
PROC PRINT DATA=Soundex_Matches NOOBS ;
   TITLE "Soundex Algorithm Matches" ;
   RUN ;
PROC SQL ;
   SELECT * FROM mydata.Movies_with_Messy_Data ;
   WHERE Title =* "Michael" ;
   QUIT ;
```

Results:

<table>
<thead>
<tr>
<th>Title</th>
<th>Length</th>
<th>Category</th>
<th>Year</th>
<th>Studio</th>
<th>Rating</th>
</tr>
</thead>
<tbody>
<tr>
<td>Michael</td>
<td>106</td>
<td>Drama</td>
<td>1997</td>
<td>Warner Bros</td>
<td>PG-13</td>
</tr>
<tr>
<td>Micheal</td>
<td>106</td>
<td>Drama</td>
<td>1997</td>
<td>Warner Brothers</td>
<td>PG-13</td>
</tr>
</tbody>
</table>

Figure 16. Soundex matches.

**EXPLORING THE SPEDIS FUNCTION**

The SPEDIS, Spelling Distance, function and its two arguments evaluates possible matching scenarios by translating a keyword into a query containing the smallest distance value. Because the SPEDIS function evaluates numerous scenarios, it can experience varying performance issues in comparison to other matching techniques. The SPEDIS function returns a value of zero when the query and keyword arguments match exactly. Users can also liberalize the matching process by specifying spelling distance values greater than zero (e.g., 10, 20, etc.). The general syntax of the SPEDIS function takes the form of:

```
SPEDIS(query, keyword)
```

In the next example, a simple DATA step with a WHERE statement and a PROC SQL with a WHERE-clause are illustrated to how the SPEDIS function is used to find exact matches for Movie Titles.

SPEDIS Function:

```plaintext
DATA SPEDIS_Matches ;
   SET mydata.Movies_with_Messy_Data ;
   WHERE SPEDIS(Title,"Michael") = 0 ;
   RUN ;
PROC SQL ;
   SELECT *
   FROM mydata.Movies_with_Messy_Data ;
   WHERE Title =* "Michael" ;
   QUIT ;
```
RUN ;
PROC PRINT DATA=work.SPEDIS_Matches NOOBS ;
TITLE “SPEDIS Function Matches” ;
RUN ;
FROM mydata.Movies_with_Messy_Data
WHERE SPEDIS(Title,“Michael”)=0 ;
QUIT ;

Results:

<table>
<thead>
<tr>
<th>Title</th>
<th>Length</th>
<th>Category</th>
<th>Year</th>
<th>Studio</th>
<th>Rating</th>
</tr>
</thead>
<tbody>
<tr>
<td>Michael</td>
<td>106</td>
<td>Drama</td>
<td>1997</td>
<td>Warner Bros</td>
<td>PG-13</td>
</tr>
</tbody>
</table>

Figure 17.  Spedis matches.

In the next example, a DATA step with a WHERE statement and a PROC SQL with a WHERE-clause are illustrated to show how the SPEDIS function is used to find spelling variations associated with Movie Titles.

SPEDIS Function:

```sas
DATA SPEDIS_Matches ;
    SET mydata.Movies_with_Messy_Data ;
    WHERE SPEDIS(Title,”Michael”) LE 20 ;
RUN ;
PROC PRINT DATA=work.SPEDIS_Matches NOOBS ;
TITLE “SPEDIS Function Matches” ;
RUN ;
PROC SQL ;
SELECT * FROM mydata.Movies_with_Messy_Data WHERE SPEDIS(Title,”Michael”) LE 20 ;
QUIT ;
```

Results:

<table>
<thead>
<tr>
<th>Title</th>
<th>Length</th>
<th>Category</th>
<th>Year</th>
<th>Studio</th>
<th>Rating</th>
</tr>
</thead>
<tbody>
<tr>
<td>Michael</td>
<td>106</td>
<td>Drama</td>
<td>1997</td>
<td>Warner Bros</td>
<td>PG-13</td>
</tr>
<tr>
<td>Micheal</td>
<td>106</td>
<td>Drama</td>
<td>1997</td>
<td>Warner Brothers</td>
<td>PG-13</td>
</tr>
</tbody>
</table>

Figure 18.  Spedis matches with equality condition loosened.

**EXPLORING THE COMPGED FUNCTION**

The COMPGED function is another fuzzy matching technique used by the SAS user community. It works by computing and using a Generalized Edit Distance (GED) score when comparing two text strings. In Teres (2011), the author describes the Generalized Edit Distance score as “a generalization of the Levenshtein edit distance, which is a measure of dissimilarity between two strings.” Sloan and Hoicowitz describe their experience using the COMPGED function to match data sets with unreliable identifiers (or keys) by pointing out, “The higher the GED score the less likely the two strings match.” Conversely, for the greatest likelihood of a match with the COMPGED function users should seek the lowest derived score from evaluating all the possible ways of matching string-1 with string-2.
The COMPGED function returns values that are multiples of 10, e.g., 20, 100, 200, etc. In Cadieux and Bretheim’s (2014) paper, the authors mention that most COMPGED scores of 100 or less are valid matches. So how is the COMPGED function used to compare two text strings for possible matching? The general syntax of the COMPGED function takes the form of:

```
COMPGED (string-1, string-2 <, cutoff-value> <, modifier)
```

**Required Arguments:**
- `string-1` specifies a character variable, constant or expression.
- `string-2` specifies a character variable, constant or expression.

**Optional Arguments:**
- `cutoff-value` specifies a numeric variable, constant or expression. If the actual generalized edit distance is greater than the value of `cutoff`, the value that is returned is equal to the value of `cutoff`.
- `modifier` specifies a value that alters the action of the COMPGED function. Valid modifier values are:
  - `i` or `I` Ignores the case in string-1 and string-2.
  - `l` or `L` Removes leading blanks before comparing the values in string-1 or string-2.
  - `n` or `N` Ignores quotation marks around string-1 or string-2.
  - `:` (colon) Truncates the longer of string-1 or string-2 to the length of the shorter string.

In this first example, a PROC SQL inner join is constructed along with the specification of a COMPGED function to allow for matches that are not perfect. The COMPGED function derives a value corresponding to the computed generalized edit distance (GED) score as, `COMPGED_Score` in the new table, `Movies_Fuzzy_Matches`. As illustrated in the results, the `COMPGED_Score` column contains a subsetted value between 0 and 100 due to the "cutoff-value" of 100 as is specified in the WHERE-clause expression. Along with the "cutoff-value, the WHERE-clause also eliminates missing titles from further consideration.

**PROC SQL Join with COMPGED Function:**

```sql
proc sql noprint ;
create table Movies_Fuzzy_Matches as
    select M.Title,
           Rating,
           Category,
           Actor_Leading,
           Actor_Supporting,
           COMPGED(M.Title,A.Title) AS COMPGED_Score
    from mydata.Movies_with_Messy_Data M,
         mydata.Actors_with_Messy_Data A
    where M.Title NE "" AND
          CALCULATED COMPGED_Score LE 100
    order by M.Title ;
quit ;
```

Results:
In the next example, the “cutoff-value” is maintained at 100, as it was in the previous example. In addition to the COMPGED function, a modifier value of “I” has been specified to tell SAS to ignore the case of both string-1 and string-2. Unlike the previous example’s results, the results for this example show that the row associated with the movie “Ghost” in the argument for string-1 matches the value of “GHOST” in the argument for string-2.

PROC SQL Join with COMPGED Function and a Modifier of ‘I’:

```sql
proc sql noprint ;
create table Movies_Fuzzy_Matches as
select M.Title,
       Rating,
       Category,
       Actor_Leading,
       Actor_Supporting,
       COMPGED(M.Title,A.Title,'I') AS COMPGED_Score
from mydata.Movies_with_Messy_Data M,
     mydata.Actors_with_Messy_Data A
where M.Title NE "" AND
     CALCAULATED COMPGED_Score LE 100
order by M.Title ;
quit ;
```

Results:
Figure 20. COMPGED output when case is considered.

In the next example, the COMPGED function’s modifier value of “I” has been removed and the “cutoff-value” was increased from 100 to 400. By increasing the “cutoff-value”, we liberalized the matching process to perform matches when the matching columns are not perfect. Unlike the previous example where the modifier value of “I” was specified, the results for this example show the row associated with the movie “Ghost” with a COMPGED_Score of 400, and the argument for string-1 matches the value of “GHOST” in the argument for string-2.

PROC SQL Join with COMPGED Function and COMPGED_Score LE 400:

```sql
proc sql noprint ;
create table Movies_Fuzzy_Matches as
select M.Title,
       Rating,
       Category,
       Actor_Leading,
       Actor_Supporting,
       COMPGED(M.Title,A.Title) AS COMPGED_Score
from mydata.Movies_with_Messy_Data M,
     mydata.Actors_with_Messy_Data A
where M.Title NE "" AND
     CALCULATED COMPGED_Score LE 400
order by M.Title ;
quit ;
```

Results:
In the next example, the COMPGED function has a “cutoff-value” for the COMPGED_Score set at 100, and a modifier value of “INL” to ignore the case, remove leading blanks, and ignore quotes around string-1 and string-2. As before, the results for this example show the row associated with the movie “Ghost” in the argument for string-1 matches the value of “GHOST” in the argument for string-2.

PROC SQL Join with COMPGED Function and Modifier of ‘INL’:

```sql
proc sql noprint;
create table Movies_Fuzzy_Matches as
select M.Title,
       Rating,
       Category,
       Actor_Leading,
       Actor_Supporting,
       COMPGED(M.Title,A.Title,'INL') AS COMPGED_Score
from mydata.Movies_with_Messy_Data M,
     mydata.Actors_with_Messy_Data A
where M.Title NE "" AND
     CALCULATED COMPGED_Score LE 100
order by M.Title;
quit;
```

Results:
EXPLORING THE COMPLEV FUNCTION

The COMPLEV function is another fuzzy matching technique used by the SAS user community. It stands for Levenshtein Edit Distance. As with the SPEDIS and COMPGED functions, the COMPLEV function provides an indicator of how close two strings are, with one exception. In lieu of assigning a score for each operation, it returns the number of operations. The general syntax of the COMPLEV function takes the form of:

```plaintext
COMPLEV( string-1, string-2 <, cutoff-value> <, modifier>)
```

**Required Arguments:**
- `string-1` specifies a character variable, constant or expression.
- `string-2` specifies a character variable, constant or expression.

**Optional Arguments:**
- `cutoff-value` specifies a numeric variable, constant or expression. If the actual Levenshtein edit distance is greater than the value of `cutoff`, the value that is returned is equal to the value of `cutoff`.
- `modifier` specifies a value that alters the action of the COMPLEV function. Valid modifier values are:
  - `i` or `I` Ignores the case in `string-1` and `string-2`.
  - `l` or `L` Removes leading blanks before comparing the values in `string-1` or `string-2`.
  - `n` or `N` Ignores quotation marks around `string-1` or `string-2`.
  - `:` (colon) Truncates the longer of `string-1` or `string-2` to the length of the shorter string.

In the next example, a PROC SQL inner join is constructed along with the specification of a COMPLEV function to determine the best possible match producing a value of `COMPLEV_Number`. As illustrated in the results, the `COMPLEV_Number` column displays the number of operations that have been performed. The lower the value the better the match (e.g., 0 = Best match, 1 = Next Best match, etc.).

**PROC SQL Join with COMPLEV Function:**

```sql
proc sql;
select M.Title,
```

![Figure 22. COMPGED output with INL specified.](image)

<table>
<thead>
<tr>
<th>Title</th>
<th>Rating</th>
<th>Category</th>
<th>Actor_Leading</th>
<th>Actor_Supporting</th>
<th>COMPGED_Score</th>
</tr>
</thead>
<tbody>
<tr>
<td>Brave Heart</td>
<td>R</td>
<td>Action Adventure</td>
<td>Mel Gibson</td>
<td>Sophie Marceau</td>
<td>0</td>
</tr>
<tr>
<td>Brave Heart</td>
<td>R</td>
<td>Action Adventure</td>
<td>Mel Gibson</td>
<td>Sophie Marceau</td>
<td>0</td>
</tr>
<tr>
<td>Coming to America</td>
<td>R</td>
<td>Comedy</td>
<td>Eddie Murphy</td>
<td>Alexei Hall</td>
<td>0</td>
</tr>
<tr>
<td>Forrest Gump</td>
<td>PG-13</td>
<td>Drama</td>
<td>Tom Hanks</td>
<td>Sally Field</td>
<td>0</td>
</tr>
<tr>
<td>Forrest Gump</td>
<td>PG-13</td>
<td>Drama</td>
<td>Tom Hanks</td>
<td>Sally Field</td>
<td>0</td>
</tr>
<tr>
<td>Ghost</td>
<td>PG-13</td>
<td>Drama Romance</td>
<td>Patrick Swayze</td>
<td>Donnie Moore</td>
<td>0</td>
</tr>
<tr>
<td>Lethal Weapon</td>
<td>R</td>
<td>Action &amp; Robber</td>
<td>Mel Gibson</td>
<td>Danny Glover</td>
<td>0</td>
</tr>
<tr>
<td>Michael</td>
<td>PG-13</td>
<td>Drama</td>
<td>John Travolta</td>
<td>Andie MacDowell</td>
<td>0</td>
</tr>
<tr>
<td>Michael</td>
<td>PG-13</td>
<td>Drama</td>
<td>John Travolta</td>
<td>Andie MacDowell</td>
<td>20</td>
</tr>
<tr>
<td>National Lampoon’s Vacation</td>
<td>PG-13</td>
<td>Comedy</td>
<td>Chevy Chase</td>
<td>Beverly D’Angelo</td>
<td>0</td>
</tr>
<tr>
<td>National Lampoon’s Vacation</td>
<td>PG-13</td>
<td>Comedy</td>
<td>Chevy Chase</td>
<td>Beverly D’Angelo</td>
<td>0</td>
</tr>
<tr>
<td>National Lampoons Vacation</td>
<td>PG-13</td>
<td>Comedy</td>
<td>Chevy Chase</td>
<td>Beverly D’Angelo</td>
<td>0</td>
</tr>
<tr>
<td>National Lampoons Vacation</td>
<td>PG-13</td>
<td>Comedy</td>
<td>Chevy Chase</td>
<td>Beverly D’Angelo</td>
<td>30</td>
</tr>
<tr>
<td>Rocky</td>
<td>PG</td>
<td>Action Adventure</td>
<td>Sylvester Stallone</td>
<td>Tala Shire</td>
<td>0</td>
</tr>
<tr>
<td>Rocky</td>
<td>PG</td>
<td>Action Adventure</td>
<td>Sylvester Stallone</td>
<td>Tala Shire</td>
<td>0</td>
</tr>
<tr>
<td>Silence of the Lambs</td>
<td>R</td>
<td>Drama Suspense</td>
<td>Anthony Hopkins</td>
<td>Jodie Foster</td>
<td>0</td>
</tr>
<tr>
<td>The Hurt for Red October</td>
<td>GP</td>
<td>Action Adventure</td>
<td>Sean Connery</td>
<td>Alec Baldwin</td>
<td>$100</td>
</tr>
<tr>
<td>The Terminator</td>
<td>R</td>
<td>Action Sci-Fi</td>
<td>Arnold Schwarzenegger</td>
<td>Michael Biehn</td>
<td>0</td>
</tr>
<tr>
<td>Titanic</td>
<td>PG-13</td>
<td>Drama Romance</td>
<td>Leonardo DiCaprio</td>
<td>Kate Winslet</td>
<td>0</td>
</tr>
</tbody>
</table>
In the next example, we modify what was produced previously and restrict our PROC SQL WHERE-clause to subset non-missing Titles and COMPLEV_Number values containing either 0 or 1. The results confirm that a fuzzy matching process using the COMPLEV function to select values of 0 or 1, representing the “best” matches for COMPLEV_Number, has been correctly performed.

PROC SQL Join with COMPLEV Function:

```sql
proc sql;
title "COMPLEV Function Matches";
select M.Title,
     Rating,
     Length,
     Category,
     COMPLEV(M.Category."Drama") as COMPLEV_Number
from mydata.Movies_with_Messy_Data M
where M.Title NE ""
order by M.Title;
quit;
```

Results:

![Figure 23. COMPLEV output.](image)

In the next example, we modify what was produced previously and restrict our PROC SQL WHERE-clause to subset non-missing Titles and COMPLEV_Number values containing either 0 or 1. The results confirm that a fuzzy matching process using the COMPLEV function to select values of 0 or 1, representing the “best” matches for COMPLEV_Number, has been correctly performed.

PROC SQL Join with COMPLEV Function:

```sql
proc sql;
title "COMPLEV Function Matches";
select M.Title,
     Rating,
     Length,
     Category,
     COMPLEV(M.Category."Drama") as COMPLEV_Number
from mydata.Movies_with_Messy_Data M
where M.Title NE ""
order by M.Title;
quit;
```
Length,
Category,

\text{COMPLEV(M.Category,"Drama") AS COMPLEV\_Number}

\text{from mydata.Movies\_with\_Messy\_Data M}
where M.Title NE ""
\text{AND COMPLEV\_Number LE 1}
order by M.Title ;
quit ;

Results:

<table>
<thead>
<tr>
<th>Title</th>
<th>Rating</th>
<th>Length</th>
<th>Category</th>
<th>COMPLEV_Number</th>
</tr>
</thead>
<tbody>
<tr>
<td>1</td>
<td>Casablanca</td>
<td>PG</td>
<td>103</td>
<td>Drama</td>
</tr>
<tr>
<td>2</td>
<td>Forrest Gump</td>
<td>PG-13</td>
<td>143</td>
<td>Drama</td>
</tr>
<tr>
<td>3</td>
<td>Forrest Gump</td>
<td>PG-13</td>
<td>142</td>
<td>Drama</td>
</tr>
<tr>
<td>4</td>
<td>Forrest Gump</td>
<td>PG-13</td>
<td>143</td>
<td>Drama</td>
</tr>
<tr>
<td>5</td>
<td>Michael</td>
<td>PG-13</td>
<td>106</td>
<td>Drama</td>
</tr>
<tr>
<td>6</td>
<td>Micheal</td>
<td>PG-13</td>
<td>106</td>
<td>Drama</td>
</tr>
</tbody>
</table>

Figure 24. COMPLEV output with additional restrictions.

**CONCLUSION**

When data originating from multiple sources contain duplicate observations, duplicate and/or unreliable keys, missing values, invalid values, capitalization and punctuation issues, inconsistent matching variables, and imprecise text identifiers, the matching process is often compromised by unreliable and/or unpredictable results. This paper demonstrates a five-step approach including identifying, cleaning and standardizing data irregularities, conducting data transformations, and utilizing special-purpose programming techniques such as the application of SAS functions, the SOUNDEX algorithm, the SPEDIS function, approximate string matching functions including COMPGED and COMPLEV, and an assortment of constructive programming techniques to standardize and combine data sets together when the matching columns are unreliable or less than perfect.

**REFERENCES**


TRADEMARK CITATIONS
SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration. Other brand and product names are trademarks of their respective companies.

ABOUT THE AUTHORS
Stephen Sloan has worked at Accenture in the Services, Consulting, and Digital groups and is currently a senior manager in the SAS Analytics area. He has worked in a variety of functional areas including Project Management, Data Management, and Statistical Analysis. Stephen has had the good fortune to have presented at many SAS conferences and to have worked with many talented people at SAS Institute. Stephen has a B.A. in Mathematics from Brandeis University, M.S. degrees in Mathematics and Computer Science from Northern Illinois University, and an MBA from Stern Business School at New York University.

Kirk Paul Lafler is an entrepreneur, consultant and founder of Software Intelligence Corporation, and has been using SAS since 1979. Kirk is a SAS Certified Professional, provider of IT consulting services, professor at UC San Diego Extension and educator to SAS users around the world, mentor, and emeritus sasCommunity.org Advisory Board member. As the author of six books including Google® Search Complete (Odyssey Press. 2014) and PROC SQL: Beyond the Basics Using SAS, Second Edition (SAS Press. 2013); Kirk has written hundreds of papers and articles; been an Invited speaker and trainer at hundreds of SAS International, regional, special-interest, local, and in-house user group conferences and meetings; and is the recipient of 25 “Best” contributed paper, hands-on workshop (HOW), and poster awards.
Your comments are valued and encouraged. Comments and suggestions may be sent to:

Stephen Sloan
Accenture
917-375-2937
Stephen.B.Sloan@accenture.com

Kirk Paul Lafler
Software Intelligence Corporation
LinkedIn: http://www.linkedin.com/in/KirkPaulLafler
Twitter: @sasNerd